

# NAG Fortran Library Routine Document

## F04CDF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F04CDF computes the solution to a complex system of linear equations  $AX = B$ , where  $A$  is an  $n$  by  $n$  Hermitian positive-definite matrix and  $X$  and  $B$  are  $n$  by  $r$  matrices. An estimate of the condition number of  $A$  and an error bound for the computed solution are also returned.

### 2 Specification

```
SUBROUTINE F04CDF (UPLO, N, NRHS, A, LDA, B, LDB, RCOND, ERRBND, IFAIL)
INTEGER          N, NRHS, LDA, LDB, IFAIL
double precision RCOND, ERRBND
complex*16      A(LDA,*), B(LDB,*)
CHARACTER*1      UPLO
```

### 3 Description

The Cholesky factorization is used to factor  $A$  as  $A = U^H U$ , if  $UPLO = 'U'$ , or  $A = LL^H$ , if  $UPLO = 'L'$ , where  $U$  is an upper triangular matrix and  $L$  is a lower triangular matrix. The factored form of  $A$  is then used to solve the system of equations  $AX = B$ .

### 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: <http://www.netlib.org/lapack/lug>

Higham N J (2002) *Accuracy and Stability of Numerical Algorithms* (2nd Edition) SIAM, Philadelphia

### 5 Parameters

- 1: UPLO – CHARACTER\*1 *Input*  
*On entry:* if UPLO = 'U', the upper triangle of the matrix  $A$  is stored, if UPLO = 'L', the lower triangle of the matrix  $A$  is stored.  
*Constraint:* UPLO = 'U' or 'L'.
- 2: N – INTEGER *Input*  
*On entry:* the number of linear equations  $n$ , i.e., the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 3: NRHS – INTEGER *Input*  
*On entry:* the number of right-hand sides  $r$ , i.e., the number of columns of the matrix  $B$ .  
*Constraint:* NRHS  $\geq 0$ .

- 4: A(LDA,\*) – **complex\*16** array *Input/Output*  
**Note:** the second dimension of the array A must be at least  $\max(1, N)$ .  
*On entry:* the  $n$  by  $n$  Hermitian matrix  $A$ .  
 If UPLO = 'U', the leading  $N$  by  $N$  upper triangular part of  $A$  contains the upper triangular part of the matrix  $A$ , and the strictly lower triangular part of  $A$  is not referenced;  
 if UPLO = 'L', the leading  $N$  by  $N$  lower triangular part of  $A$  contains the lower triangular part of the matrix  $A$ , and the strictly upper triangular part of  $A$  is not referenced.  
*On exit:* if IFAIL = 0 or  $N + 1$ , the factor  $U$  or  $L$  from the Cholesky factorization  $A = U^H U$  or  $A = L L^H$ .
- 5: LDA – INTEGER *Input*  
*On entry:* the first dimension of the array A as declared in the (sub)program from which F04CDF is called.  
*Constraint:*  $LDA \geq \max(1, N)$ .
- 6: B(LDB,\*) – **complex\*16** array *Input/Output*  
**Note:** the second dimension of the array B must be at least  $\max(1, NRHS)$ . To solve the equations  $Ax = b$ , where  $b$  is a single right-hand side, B may be supplied as a one-dimensional array with length  $LDB = \max(1, N)$ .  
*On entry:* the  $n$  by  $r$  matrix of right-hand sides  $B$ .  
*On exit:* if IFAIL = 0 or  $N + 1$ , the  $n$  by  $r$  solution matrix  $X$ .
- 7: LDB – INTEGER *Input*  
*On entry:* the first dimension of the array B as declared in the (sub)program from which F04CDF is called.  
*Constraint:*  $LDB \geq \max(1, N)$ .
- 8: RCOND – **double precision** *Output*  
*On exit:* if IFAIL = 0 or  $N + 1$ , an estimate of the reciprocal of the condition number of the matrix  $A$ , computed as  $RCOND = 1 / (\|A\|_1 \|A^{-1}\|_1)$ .
- 9: ERRBND – **double precision** *Output*  
*On exit:* if IFAIL = 0 or  $N + 1$ , an estimate of the forward error bound for a computed solution  $\hat{x}$ , such that  $\|\hat{x} - x\|_1 / \|x\|_1 \leq ERRBND$ , where  $\hat{x}$  is a column of the computed solution returned in the array B and  $x$  is the corresponding column of the exact solution  $X$ . If RCOND is less than **machine precision**, then ERRBND is returned as unity.
- 10: IFAIL – INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.  
*On exit:* IFAIL = 0 unless the routine detects an error (see Section 6).  
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

## 6 Error Indicators and Warnings

If on entry  $\text{IFAIL} = 0$  or  $-1$ , explanatory error messages are output on the current error message unit (as defined by  $\text{X04AAF}$ ).

Errors or warnings detected by the routine:

$\text{IFAIL} < 0$  and  $\text{IFAIL} \neq -999$

If  $\text{IFAIL} = -i$ , the  $i$ th argument had an illegal value.

$\text{IFAIL} = -999$

Allocation of memory failed. The **double precision** allocatable memory required is  $N$ , and the **complex\*16** allocatable memory required is  $2 \times N$ . Allocation failed before the solution could be computed.

$\text{IFAIL} > 0$  and  $\text{IFAIL} \leq N$

If  $\text{IFAIL} = i$ , the leading minor of order  $i$  of  $A$  is not positive-definite. The factorization could not be completed, and the solution has not been computed.

$\text{IFAIL} = N + 1$

$\text{RCOND}$  is less than **machine precision**, so that the matrix  $A$  is numerically singular. A solution to the equations  $AX = B$  has nevertheless been computed.

## 7 Accuracy

The computed solution for a single right-hand side,  $\hat{x}$ , satisfies an equation of the form

$$(A + E)\hat{x} = b,$$

where

$$\|E\|_1 = O(\epsilon)\|A\|_1$$

and  $\epsilon$  is the **machine precision**. An approximate error bound for the computed solution is given by

$$\frac{\|\hat{x} - x\|_1}{\|x\|_1} \leq \kappa(A) \frac{\|E\|_1}{\|A\|_1},$$

where  $\kappa(A) = \|A^{-1}\|_1 \|A\|_1$ , the condition number of  $A$  with respect to the solution of the linear equations. F04CDF uses the approximation  $\|E\|_1 = \epsilon \|A\|_1$  to estimate  $\text{ERRBND}$ . See Section 4.4 of Anderson *et al.* (1999) for further details.

## 8 Further Comments

The total number of floating-point operations required to solve the equations  $AX = B$  is proportional to  $(\frac{1}{3}n^3 + n^2r)$ . The condition number estimation typically requires between four and five solves and never more than eleven solves, following the factorization.

In practice the condition number estimator is very reliable, but it can underestimate the true condition number; see Section 15.3 of Higham (2002) for further details.

The real analogue of F04CDF is F04BDF.

## 9 Example

To solve the equations

$$AX = B,$$

where  $A$  is the Hermitian positive-definite matrix

$$A = \begin{pmatrix} 3.23 & 1.51 - 1.92i & 1.90 + 0.84i & 0.42 + 2.50i \\ 1.51 + 1.92i & 3.58 & -0.23 + 1.11i & -1.18 + 1.37i \\ 1.90 - 0.84i & -0.23 - 1.11i & 4.09 & 2.33 - 0.14i \\ 0.42 - 2.50i & -1.18 - 1.37i & 2.33 + 0.14i & 4.29 \end{pmatrix}$$

and

$$B = \begin{pmatrix} 3.93 - 6.14i & 1.48 + 6.58i \\ 6.17 + 9.42i & 4.65 - 4.75i \\ -7.17 - 21.83i & -4.91 + 2.29i \\ 1.99 - 14.38i & 7.64 - 10.79i \end{pmatrix}.$$

An estimate of the condition number of  $A$  and an approximate error bound for the computed solutions are also printed.

### 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F04CDF Example Program Text
*      Mark 21 Release. NAG Copyright 2004.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, NRHSMX
      PARAMETER        (NMAX=8,NRHSMX=2)
      INTEGER          LDA, LDB
      PARAMETER        (LDA=NMAX,LDB=NMAX)
*      .. Local Scalars ..
      DOUBLE PRECISION ERRBND, RCOND
      INTEGER          I, IERR, IFAIL, J, N, NRHS
*      .. Local Arrays ..
      COMPLEX *16      A(LDA,NMAX), B(LDB,NRHSMX)
      CHARACTER        CLABS(1), RLABS(1)
*      .. External Subroutines ..
      EXTERNAL         F04CDF, X04DBF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F04CDF Example Program Results'
      WRITE (NOUT,*)
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, NRHS
      IF (N.LE.NMAX .AND. NRHS.LE.NRHSMX) THEN
*
*         Read the upper triangular part of A from data file
*
*         READ (NIN,*) ((A(I,J),J=I,N),I=1,N)
*
*         Read B from data file
*
*         READ (NIN,*) ((B(I,J),J=1,NRHS),I=1,N)
*
*         Solve the equations AX = B for X
*
*         IFAIL = -1
*         CALL F04CDF('Upper',N,NRHS,A,LDA,B,LDB,RCOND,ERRBND,IFAIL)
*
*         IF (IFAIL.EQ.0) THEN
```

```

*          Print solution, estimate of condition number and approximate
*          error bound
*
          IERR = 0
          CALL X04DBF('General',' ',N,NRHS,B,LDB,'Bracketed',' ',
+                    'Solution','Integer',RLABS,'Integer',CLABS,80,0,
+                    IERR)
*
          WRITE (NOUT,*)
          WRITE (NOUT,*) 'Estimate of condition number'
          WRITE (NOUT,99999) 1.0D0/RCOND
          WRITE (NOUT,*)
          WRITE (NOUT,*)
+          'Estimate of error bound for computed solutions'
          WRITE (NOUT,99999) ERRBND
          ELSE IF (IFAIL.EQ.N+1) THEN
*
*          Matrix A is numerically singular. Print estimate of
*          reciprocal of condition number and solution
*
          WRITE (NOUT,*)
          WRITE (NOUT,*) 'Estimate of reciprocal of condition number'
          WRITE (NOUT,99999) RCOND
*
          WRITE (NOUT,*)
          IERR = 0
          CALL X04DBF('General',' ',N,NRHS,B,LDB,'Bracketed',' ',
+                    'Solution','Integer',RLABS,'Integer',CLABS,80,0,
+                    IERR)
*
          ELSE IF (IFAIL.GT.0 .AND. IFAIL.LE.N) THEN
*
*          The matrix A is not positive definite to working precision
*
          WRITE (NOUT,99998) 'The leading minor of order ', IFAIL,
+          ' is not positive definite'
          END IF
          ELSE
          WRITE (NOUT,*) 'NMAX and/or NRHSMX too small'
          END IF
          STOP
*
99999 FORMAT (8X,1P,E9.1)
99998 FORMAT (1X,A,I3,A)
          END

```

## 9.2 Program Data

F04CDF Example Program Data

```

      4              2                               :Values of N and NRHS

( 3.23,  0.00) ( 1.51, -1.92) ( 1.90,  0.84) ( 0.42,  2.50)
              ( 3.58,  0.00) (-0.23,  1.11) (-1.18,  1.37)
                                ( 4.09,  0.00) ( 2.33, -0.14)
                                ( 4.29,  0.00) :End of matrix A

( 3.93, -6.14) ( 1.48,  6.58)
( 6.17,  9.42) ( 4.65, -4.75)
(-7.17,-21.83) (-4.91,  2.29)
( 1.99,-14.38) ( 7.64,-10.79)                               :End of matrix B

```

### 9.3 Program Results

F04CDF Example Program Results

Solution

			1			2
1	(	1.0000,	-1.0000)	(	-1.0000,	2.0000)
2	(	-0.0000,	3.0000)	(	3.0000,	-4.0000)
3	(	-4.0000,	-5.0000)	(	-2.0000,	3.0000)
4	(	2.0000,	1.0000)	(	4.0000,	-5.0000)

Estimate of condition number  
1.5E+02

Estimate of error bound for computed solutions  
1.7E-14

---